// File: contracts/interfaces/IPowerswapFactory.sol

pragma solidity ^0.7.4;

interface IPowerswapFactory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function feeTo() external view returns (address);
    function feeToSetter() external view returns (address);
    function rSetter() external view returns (address);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function setFeeTo(address) external;
    function setFeeToSetter(address) external;
    function setrSetter(address) external;
}

// File: contracts/interfaces/IPowerswapERC20.sol

//pragma solidity = 0.5.16

interface IPowerswapERC20 {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);
    function symbol() external pure returns (string memory);
    function decimals() external pure returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function DOMAIN_SEPARATOR() external view returns (bytes32);
    function PERMIT_TYPEHASH() external pure returns (bytes32);
    function nonces(address owner) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external;
}

// File: contracts/interfaces/IPowerswapPair.sol


interface IPowerswapPair is IPowerswapERC20{

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(address indexed sender, uint amount0, uint amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint);
    function factory() external view returns (address);
    function token0() external view returns (address);
    function token1() external view returns (address);
    function getReserves() external view returns (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast);
    function getBalances() external view returns(uint _balanceBuy0, uint _balanceBuy1, uint _balanceSell0, uint _balanceSell1);
    function getPrices() external view returns(uint _priceBuy, uint _priceSell);
    function setR(uint16) external;
    function getR() external view returns(uint16);
    function price0CumulativeLast() external view returns (uint);
    function price1CumulativeLast() external view returns (uint);
    function kLast() external view returns (uint);

    function mint(address to) external returns (uint liquidity);
    function burn(address to) external returns (uint amount0, uint amount1);
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external;
    function updateVirtualBalances() external;
    function skim(address to) external;
    function sync() external;

    function initialize(address, address) external;
}

// File: contracts/libraries/SafeMath.sol


library SafeMath{
    function add(uint x, uint y) internal pure returns(uint z){
        require((z = x + y) >= x, 'ds-math-add-overflow');
    }
    
    function sub(uint x, uint y) internal pure returns (uint z) {
        require((z = x - y) <= x, 'ds-math-sub-underflow');
    }

    function mul(uint x, uint y) internal pure returns (uint z) {
        require(y == 0 || (z = x * y) / y == x, 'ds-math-mul-overflow');
    }
}

// File: contracts/PowerswapERC20.sol


abstract contract PowerswapERC20 is IPowerswapERC20{
    using SafeMath for uint;
    
    string public constant override name = 'Powerswap';
    string public constant override symbol = 'POWER';
    uint8 public constant override decimals = 18;
    uint public override totalSupply;
    mapping(address => uint) public override balanceOf;
    mapping(address => mapping(address => uint)) public override allowance;
    
    bytes32 public override DOMAIN_SEPARATOR;
    // keccak256("Permit(address owner,address spender,uint256 value,uint256 nonce,uint256 deadline)");
    bytes32 public constant override PERMIT_TYPEHASH = 0x6e71edae12b1b97f4d1f60370fef10105fa2faae0126114a169c64845d6126c9;
    mapping(address => uint) public override nonces; 
    
//    event Approval(address indexed owner, address indexed spender, uint value);
//    event Transfer(address indexed from, address indexed to, uint value);

    constructor() {
        uint chainId;
        assembly {
            chainId := chainid()
        }
        DOMAIN_SEPARATOR = keccak256(
            abi.encode(
                keccak256('EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)'),
                keccak256(bytes(name)),
                keccak256(bytes('1')),
                chainId,
                address(this)
            )
        );
    }

    
    function _mint(address to, uint value) internal {
        totalSupply = totalSupply.add(value);
        balanceOf[to] = balanceOf[to].add(value);
        emit Transfer(address(0), to, value);
    }
    
    function _burn(address from, uint value) internal{
        balanceOf[from] = balanceOf[from].sub(value);
        totalSupply = totalSupply.sub(value);
        emit Transfer(from, address(0), value);
    }
    
    function _approve(address owner, address spender, uint value) private{
        allowance[owner][spender] = value;
        emit Approval(owner, spender, value);
    }
    
    function _transfer(address from, address to, uint value) private{
        balanceOf[from] = balanceOf[from].sub(value);
        balanceOf[to] = balanceOf[to].add(value);
        emit Transfer(from, to, value);
    }
    
    function approve(address spender, uint value) external override returns(bool){
        _approve(msg.sender, spender, value);
        return true;
    }
    
    function transfer(address to, uint value) external override returns(bool){
    _transfer(msg.sender, to, value);
    return true;
    }    
    
    function transferFrom(address from, address to, uint value) external override returns(bool){
        if(allowance[from][msg.sender] != uint(-1)){
            allowance[from][msg.sender] = allowance[from][msg.sender].sub(value);
        }
        _transfer(from, to, value);
        return true;
    }
    
    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external override{
        require(deadline >= block.timestamp, 'Powerswap: EXPIRED');
        bytes32 digest = keccak256(
            abi.encodePacked(
                '\x19\x01',
                DOMAIN_SEPARATOR,
                keccak256(abi.encode(PERMIT_TYPEHASH, owner, spender, value, nonces[owner]++, deadline))
                )
            );
        address recoveredAddress = ecrecover(digest, v, r, s);
        require(recoveredAddress != address(0) && recoveredAddress == owner, 'Powerswap: INVALID_SIGNATURE');
        _approve(owner, spender, value);
    }
}

// File: contracts/libraries/Math.sol



// a library for performing various math operations

library Math {
    function min(uint x, uint y) internal pure returns (uint z) {
        z = x < y ? x : y;
    }
    
    function max(uint x, uint y) internal pure returns (uint z) {
        return x >= y ? x : y;
    }

    // babylonian method (https://en.wikipedia.org/wiki/Methods_of_computing_square_roots#Babylonian_method)
    function sqrt(uint y) internal pure returns (uint z) {
        if (y > 3) {
            z = y;
            uint x = y / 2 + 1;
            while (x < z) {
                z = x;
                x = (y / x + x) / 2;
            }
        } else if (y != 0) {
            z = 1;
        }
    }
    
        // cbrt = cube root
   function cbrt(uint y) internal pure returns(uint z){
       if(y > 7){
           z = y;
           uint x = y/3 + 1;
           while(x < z){
               z = x;
               x = (2*x + y/(x*x))/3;
           }
       }else if(y != 0){
           z = 1;
       }
   }

}

// File: contracts/libraries/UQ112x112.sol



// a library for handling binary fixed point numbers (https://en.wikipedia.org/wiki/Q_(number_format))

// range: [0, 2**112 - 1]
// resolution: 1 / 2**112

library UQ112x112 {
    uint224 constant Q112 = 2**112;

    // encode a uint112 as a UQ112x112
    function encode(uint112 y) internal pure returns (uint224 z) {
        z = uint224(y) * Q112; // never overflows.
    }

    // divide a UQ112x112 by a uint112, returning a UQ112x112
    function uqdiv(uint224 x, uint112 y) internal pure returns (uint224 z) {
        z = x / uint224(y);
    }
}

// File: contracts/interfaces/IERC20.sol

interface IERC20 {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external view returns (string memory);
    function symbol() external view returns (string memory);
    function decimals() external view returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);
}

// File: contracts/interfaces/IPowerswapCallee.sol



interface IPowerswapCallee {
    function powerswapCall(address sender, uint amount0, uint amount1, bytes calldata data) external;
}

// File: contracts/PowerswapPair.sol









contract PowerswapPair is IPowerswapPair, PowerswapERC20{
    using SafeMath for uint;
    using SafeMath for uint112;
    using UQ112x112 for uint224;
    uint8 private constant RESOLUTION = 112;

    
    uint public constant override MINIMUM_LIQUIDITY = 1000;
    bytes4 private constant SELECTOR = bytes4(keccak256(bytes('transfer(address,uint256)')));
    
    address public override factory;
    address public override token0; // 0 for 'X', 1 for 'Y';
    address public override token1;
    uint16  private  R; // R = r*(2**16-1) = r*65535; r = R/(2**16-1); rounding error
    
    uint112 private reserve0; // uses single storage slot, accessible via getReserves
    uint112 private reserve1;
    
    uint private balanceBuy0; // Buy means "Buy 1 & Sell 0";  Sell means "Sell 1 & Buy 0";
    uint private balanceBuy1;
    uint private balanceSell0; 
    uint private balanceSell1;
    
    uint private priceBuy; // P_{1/0} 
    uint private priceSell; //P_{0/1}

    uint32 private blockTimestampLast;
    
    uint public override price0CumulativeLast;
    uint public override price1CumulativeLast;
    uint public override kLast; // reserveReal0 * reserveReal1, as of immediately after the most recent liquidity event 
    
    uint private unlocked = 1;
    modifier lock(){
        require(unlocked == 1, 'Powerswap: LOCKED');
        unlocked = 0;
        _;
        unlocked = 1;
    }
    
    function getReserves() public view override returns(uint112 _reserve0, uint112 _reserve1, uint32 _blockTimestampLast){
        _reserve0 = reserve0;
        _reserve1 = reserve1;
        _blockTimestampLast = blockTimestampLast;
    }
    
    function getBalances() public view override returns(uint _balanceBuy0, uint _balanceBuy1, uint _balanceSell0, uint _balanceSell1){
        _balanceBuy0  = balanceBuy0;
        _balanceBuy1  = balanceBuy1;
        _balanceSell0 = balanceSell0;
        _balanceSell1 = balanceSell1;
    }
    
    function getPrices() public view override returns(uint _priceBuy, uint _priceSell){
        _priceBuy   = priceBuy;
        _priceSell  = priceSell;
    }
    
    function getR() public view override returns(uint16 _R){
        _R = R;
    }

    function _safeTransfer(address token, address to, uint value) private{
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(SELECTOR, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'Powerswap: TRANSFER_FAILED');
    }

    constructor() public{
        factory = msg.sender;
    } 

    // called once by the factory at time of deployment
    function initialize(address _token0, address _token1) external override{
        require(msg.sender == factory, 'Powerswap: FORBIDDEN');
        token0 = _token0;
        token1 = _token1;
        R = 32767;  // default r = 0.5;
    }

    function setR(uint16 _R) external override {
        address _rSetter = IPowerswapFactory(factory).rSetter();
        require(msg.sender == _rSetter, 'Powerswap: FORBIDDEN');
        R = _R;
    }

    //update reserves and, on the first call per block, price accumulators
    function _updateAtTheInitialBlock(uint balance0, uint balance1) private{
        require(balance0 <= uint112(-1) && balance1 <= uint112(-1), 'Powerswap: OVERFLOW');
        reserve0 = uint112(balance0);
        reserve1 = uint112(balance1);

        balanceBuy0  = balance0;
        balanceBuy1  = balance1;
        balanceSell0 = balance0;
        balanceSell1 = balance1;

        priceBuy     = (balance0 << RESOLUTION) / balance1; // original price * 2**112
        priceSell    = (balance1 << RESOLUTION) / balance0;
        
        blockTimestampLast = uint32(block.timestamp % 2**32);

        emit Sync(reserve0, reserve1);
    }

    function _updateAtTheFirstCallInEachBlock(uint112 _reserve0, uint112 _reserve1, uint32 timeElapsed) private{
        // * never overflow, and + overflow is desired
        balanceBuy0  = _reserve0;
        balanceBuy1  = _reserve1;
        balanceSell0 = _reserve0;
        balanceSell1 = _reserve1;
        
        uint _priceBuy  = priceBuy;
        uint _priceSell = priceSell;
        uint16 _R = R;
        priceBuy  = (_priceBuy.mul(_reserve1).mul(_R).add((uint(_reserve0) << RESOLUTION).mul(65535-_R)))/(_reserve1.mul(65535));
        priceSell = (_priceSell.mul(_reserve0).mul(_R).add((uint(_reserve1) << RESOLUTION).mul(65535-_R)))/(_reserve0.mul(65535));

        price0CumulativeLast += uint(UQ112x112.encode(_reserve1).uqdiv(_reserve0)) * timeElapsed;
        price1CumulativeLast += uint(UQ112x112.encode(_reserve0).uqdiv(_reserve1)) * timeElapsed;
    }

    function _update(uint balance0, uint balance1, uint112 _reserve0, uint112 _reserve1) private{
        require(balance0 <= uint112(-1) && balance1 <= uint112(-1), 'Powerswap: OVERFLOW');
        reserve0 = uint112(balance0);
        reserve1 = uint112(balance1);
        uint32 blockTimestamp = uint32(block.timestamp % 2**32);
        uint32 timeElapsed = blockTimestamp - blockTimestampLast; //overflow is desired.
        if(timeElapsed > 0 && _reserve0 != 0 && _reserve1 != 0){
            _updateAtTheFirstCallInEachBlock(reserve0, reserve1, timeElapsed);
            blockTimestampLast = blockTimestamp;
        }else{
        uint _balanceBuy0  = balanceBuy0;
        uint _balanceBuy1  = balanceBuy1;
        uint _balanceSell0 = balanceSell0;
        uint _balanceSell1 = balanceSell1;
        
        balanceBuy0     = _balanceBuy0.add(balance0).sub(_reserve0); 
        balanceBuy1     = _balanceBuy1.add(balance1).sub(_reserve1);
        balanceSell0    = _balanceSell0.add(balance0).sub(_reserve0);
        balanceSell1    = _balanceSell1.add(balance1).sub(_reserve1);
        }

        emit Sync(reserve0, reserve1);
    }
    

    function _updateAfterSwap(
        uint amount0In,  uint amount1In,
        uint amount0Out, uint amount1Out,
        uint balance0,   uint balance1) private {
/*
        uint32 blockTimestamp = uint32(block.timestamp % 2**32);
        uint32 timeElapsed = blockTimestamp - blockTimestampLast; //overflow is desired.
        if(timeElapsed > 0 && _reserve0 != 0 && _reserve1 != 0){
        _updateAtTheFirstCallInEachBlock(_reserve0, _reserve1, timeElapsed);
        blockTimestampLast = blockTimestamp;
        }
*/
        if(amount0In > 0){
        balanceBuy0 += amount0In;
        balanceBuy1 -= amount1Out;
        }else{
        balanceSell0 -= amount0Out;
        balanceSell1 += amount1In;
        }
        reserve0 = uint112(balance0);
        reserve1 = uint112(balance1);

        emit Sync(reserve0, reserve1);
    }

// if fee is on, mint liquidity equivalent to 1/6th of the growth in sqrt(k)
    function _mintFee(uint112 _reserve0, uint112 _reserve1) private returns (bool feeOn){
        address feeTo = IPowerswapFactory(factory).feeTo(); 
        feeOn = feeTo != address(0);
        uint _kLast = kLast; // gas savings
        if (feeOn) {
            if (_kLast != 0) {
                uint rootK = Math.sqrt(uint(_reserve0).mul(_reserve1));
                uint rootKLast = Math.sqrt(_kLast);
                if (rootK > rootKLast) {
                    uint numerator = totalSupply.mul(rootK.sub(rootKLast));
                    uint denominator = rootK.mul(5).add(rootKLast);
                    uint liquidity = numerator / denominator;
                    if (liquidity > 0) _mint(feeTo, liquidity);
                }
            }
        } else if (_kLast != 0) {
            kLast = 0;
        }
    }

    
    // this low-level function should be called from a contract which performs important safety checks
    function mint(address to) external lock override returns(uint liquidity){
        (uint112 _reserve0, uint112 _reserve1, ) = getReserves(); //gas savings
        uint balance0 = IERC20(token0).balanceOf(address(this));
        uint balance1 = IERC20(token1).balanceOf(address(this));
        uint amount0  = balance0.sub(_reserve0);
        uint amount1  = balance1.sub(_reserve1);
        
        bool feeOn = _mintFee(_reserve0, _reserve1);
        uint _totalSupply = totalSupply; // gas savings, must be defined here since totalSupply can update in _mintFee
        if(_totalSupply == 0){
            liquidity = Math.sqrt(amount0.mul(amount1)).sub(MINIMUM_LIQUIDITY);
            _mint(address(0), MINIMUM_LIQUIDITY); // permanently lock the first MINIMUM_LIQUIDITY tokens
            _updateAtTheInitialBlock(balance0, balance1);
        }else{
            _update(balance0, balance1, _reserve0, _reserve1);
            liquidity = Math.min(_totalSupply.mul(amount0.mul(98))/(_reserve0.mul(100).add(amount0.mul(2))),
                                 _totalSupply.mul(amount1.mul(98))/(_reserve1.mul(100).add(amount1.mul(2)))); // F = 0.98
            }
        require(liquidity > 0, 'Powerswap: INSUFFICIENT_LIQUIDITY_MINTED');
        _mint(to, liquidity);
        if(feeOn) {
        kLast = uint(_reserve0).mul(_reserve1);
        } // reserve0 and reserve1 are up-to-date
        emit Mint(msg.sender, amount0, amount1);
    }
    
    // this low-level function should be called from a contract which performs important safety checks
    function burn(address to) external lock override returns(uint amount0, uint amount1){
        (uint112 _reserve0, uint112 _reserve1, ) = getReserves(); //gas savings
        address _token0 = token0;
        address _token1 = token1;
        uint balance0 = IERC20(_token0).balanceOf(address(this)); // this = pair
        uint balance1 = IERC20(_token1).balanceOf(address(this));
        uint liquidity = balanceOf[address(this)];
        
        bool feeOn = _mintFee(_reserve0, _reserve1);
        uint _totalSupply = totalSupply; // gas savings, must be defined here since totalSupply can update in _mintFee
        uint numerator = _totalSupply.mul(100).sub(liquidity.mul(2));
        uint denominator = (_totalSupply.mul(_totalSupply).mul(98)/liquidity).add(1);
        amount0 = balance0.mul(numerator)/denominator; // F=0.98 here
        amount1 = balance1.mul(numerator)/denominator;
        require(amount0 > 0 && amount1 > 0, 'Powerswap: INSUFFICIENT_LIQUIDITY_BURNED');
        _burn(address(this), liquidity);
        _safeTransfer(_token0, to, amount0);
        _safeTransfer(_token1, to, amount1);
        balance0 = IERC20(token0).balanceOf(address(this));
        balance1 = IERC20(token1).balanceOf(address(this));
        
        _update(balance0, balance1, _reserve0, _reserve1);
        if (feeOn){
        kLast = uint(_reserve0).mul(_reserve1);
        } // reserve0 and reserve1 are up-to-date // reserve0 and reserve1 are up-to-date
        emit Burn(msg.sender, amount0, amount1, to);
    } 
     
    // this low-level function should be called from a contract which performs important safety checks
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external lock override{
        require(amount0Out > 0 || amount1Out > 0, 'Powerswap: INSUFFICIENT_OUTPUT_AMOUNT');
        (uint112 _reserve0, uint112 _reserve1,) = getReserves(); //gas savings
        require(amount0Out < _reserve0 && amount1Out < _reserve1, 'Powerswap: INSUFFICIENT_LIQUIDITY');
        
        uint balance0;
        uint balance1;
        { // scope for _token{0,1}, avoids stack too deep errors
        address _token0 = token0;
        address _token1 = token1;
        require(to != _token0 && to != _token1, 'Powerswap: INVALID_TO');
        if (amount0Out > 0) _safeTransfer(_token0, to, amount0Out); // optimistically transfer tokens
        if (amount1Out > 0) _safeTransfer(_token1, to, amount1Out); // optimistically transfer tokens
        if (data.length > 0){IPowerswapCallee(to).powerswapCall(msg.sender, amount0Out, amount1Out, data);}
        balance0 = IERC20(_token0).balanceOf(address(this));
        balance1 = IERC20(_token1).balanceOf(address(this));
        }
        uint amount0In = balance0 > _reserve0 - amount0Out ? balance0 - (_reserve0 - amount0Out) : 0;
        uint amount1In = balance1 > _reserve1 - amount1Out ? balance1 - (_reserve1 - amount1Out) : 0;
        require(amount0In > 0 || amount1In > 0, 'Powerswap: INSUFFICIENT_INPUT_AMOUNT');
        require(amount0In <= uint112(-1) && amount1In <= uint112(-1), 'Powerswap: OVERFLOW');
        _updateAfterSwap(amount0In, amount1In, amount0Out, amount1Out, balance0, balance1);
        emit Swap(msg.sender, amount0In, amount1In, amount0Out, amount1Out, to);    
    }
    
    function updateVirtualBalances() external lock override{
        uint32 blockTimestamp = uint32(block.timestamp % 2**32);
        uint32 timeElapsed = blockTimestamp - blockTimestampLast; //overflow is desired.
        uint112 _reserve0 = reserve0;
        uint112 _reserve1 = reserve1;
        if(timeElapsed > 0 && _reserve0 != 0 && _reserve1 != 0){
            _updateAtTheFirstCallInEachBlock(_reserve0, _reserve1, timeElapsed);
            blockTimestampLast = blockTimestamp;
        }
    }
    
    // force balances to match reserves 
    function skim(address to) external lock override{
        address _token0 = token0;
        address _token1 = token1;

        _safeTransfer(_token0, to, IERC20(_token0).balanceOf(address(this)).sub(reserve0));
        _safeTransfer(_token1, to, IERC20(_token1).balanceOf(address(this)).sub(reserve1));
    }
    
    // force reserves to match balances
    function sync() external lock override{
        _update(IERC20(token0).balanceOf(address(this)), IERC20(token1).balanceOf(address(this)), reserve0, reserve1);
    }
}

// File: contracts/PowerswapFactory.sol




contract PowerswapFactory is IPowerswapFactory{
    address public override feeTo;
    address public override feeToSetter;
    address public override rSetter;

    mapping(address => mapping(address => address)) public override getPair;
    address[] public override allPairs;

    constructor(address _feeToSetter, address _rSetter) public{
        feeToSetter = _feeToSetter;
        rSetter = _rSetter;
    }
    
    function allPairsLength() external view override returns(uint){
        return allPairs.length;
    }
    
    function createPair(address tokenA, address tokenB) external override returns(address pair){
        require(tokenA != tokenB, 'Powerswap: IDENTICAL_ADDRESSES');
        (address token0, address token1) = tokenA < tokenB ? (tokenA, tokenB) : (tokenB, tokenA);
        require(token0 != address(0), 'Powerswap: ZERO_ADDRESSES');
        require(getPair[token0][token1] == address(0), 'Powerswap: PAIR_EXISTS'); // single check is sufficient
        bytes memory bytecode = type(PowerswapPair).creationCode;
        bytes32 salt = keccak256(abi.encodePacked(token0, token1));
        assembly{
            pair := create2(0, add(bytecode, 32), mload(bytecode), salt) 
        }
        IPowerswapPair(pair).initialize(token0, token1);
        getPair[token0][token1] = pair;
        getPair[token1][token0] = pair;
        allPairs.push(pair);
        emit PairCreated(token0, token1, pair, allPairs.length);
    }
    
    function setFeeTo(address _feeTo) external override {
        require(msg.sender == feeToSetter, 'Powerswap: FORBIDDEN');
        feeTo = _feeTo;
    }
    
    function setFeeToSetter(address _feeToSetter) external override{
        require(msg.sender == feeToSetter, 'Powerswap: FORBIDDEN');
        feeToSetter = _feeToSetter;
    }

    function setrSetter(address _rSetter) external override{
        require(msg.sender == rSetter, 'Powerswap: FORBIDDEN');
        rSetter = _rSetter;
    }
}